



Don't Be a SAS® Dinosaur: Modernize Your SAS Programs

by Warren Repole

Frequency Reports: Sequence by Number of Occurrences


Scenario:

You want to generate a report of frequency counts, emphasizing the most common values.

The FREQ procedure can generate the counts, but the report shows the category values in numeric or alphabetical order.

The old way: Sort the PROC FREQ Output Data Set

The original approach is to sort the output data set from PROC FREQ and generate the final frequency report with PROC PRINT.



```
proc freq data=sashelp.class;
  tables sex / noprint out=genders;
  tables age / noprint out=ages;
run;
proc sort data=genders;
  by descending Count ;
run;
proc print data=genders label;
  title1 "Frequency Counts (most common to least common)";
run;
proc sort data=ages;
  by descending Count ;
run;
data ages;
  set ages;
  CumulativeFrequency+Count;
  CumulativePercent+Percent;
run;
proc print data=ages label;
  title1 "Frequency Counts (most common to least common)";
  title2 "with cumulative results";
run;
```

Please direct all correspondence to:

Warren Repole, 1705 Palm Springs Dr, Vienna VA 22182-2331

sasdinosaur@repole.com

www.repole.com/dinosaur

The new way: Use the ORDER=FREQ Option (available in SAS Release 82)

An alternate approach is to specify the ORDER=FREQ option of PROC FREQ.

```
proc freq data=sashelp.class order=freq;
  tables age sex ;
  title1 "Frequency Counts (most common to least common)";
  title2 "with cumulative results";
run;
```

Advantages of the alternate approach:

- Only one pass of the data occurs.
- Only one program step is required.
- By default, missing values are excluded as valid levels, although a notation about missing values appears in the report.

Disadvantages of the alternate approach:

- All category levels appear in the report. The original approach permits the display of a limited number of levels.
- The rank of the levels is not readily apparent. The original approach shows the rank of each level as the observation number.
- An additional PROC FREQ option is needed to include missing values as valid levels.

Additional documentation for this technique can be found in *Base SAS® 9.2 Procedures Guide. Cary, NC: SAS Institute Inc.*

Visit <http://support.sas.com/documentation/onlinedoc/sas9doc.html> for SAS 9 documentation.

Go to <http://www.repole.com/dinosaur/freqorder.html> for the sample code and output for this topic.

These techniques are mentioned in other SAS references and publications:

SAS Technical Support Document TS-407 (<http://support.sas.com/techsup/technote/ts407.html>)

SAS Sample 30867 (<http://support.sas.com/kb/30/867.html>)

SUGI Paper 252-31 (<http://www2.sas.com/proceedings/sugi31/252-31.pdf>)